



# Application Note

E 系列 EtherCAT 驅動器  
搭配 ACS SPiiPlus MMI Application Studio

## 修訂紀錄

手冊版次資訊亦標記於手冊封面右下角。

MD57UC01-2512\_V1.0

手冊版次

發行年份與月份

| 發行日期       | 版次  | 適用產品                    | 更新內容  |
|------------|-----|-------------------------|-------|
| 2025/12/17 | 1.0 | E 系列<br>EtherCAT<br>驅動器 | 初版發行。 |

## 相關文件

透過相關文件，使用者可快速了解此手冊的定位，以及各手冊、產品之間的關聯性。詳細內容請至本公司官網→下載中心→手冊總覽閱覽 ( [https://www.hiwinmikro.tw/Downloads/ManualOverview\\_TC.htm](https://www.hiwinmikro.tw/Downloads/ManualOverview_TC.htm) )。

## 序言

本手冊說明 E 系列 EtherCAT 驅動器搭配 ACS 控制器時，軟體 SPiiPlus MMI Application Studio 的操作。手冊將依照完整的設備設置來編排，依序為控制器的專案建立、通訊設定、參數設定、試運轉。欲瞭解 E 系列驅動器的詳細資訊，請參閱相關的使用者手冊。

## 軟硬體規格

| 名稱                 | 軟體 / 韌體版本                                                                                               |
|--------------------|---------------------------------------------------------------------------------------------------------|
| E1 系列 EtherCAT 驅動器 | 軟體 ( Thunder ) : 1.13.7.0 ( 含 ) 以上<br>韌體 : 2.13.6 ( 含 ) 以上<br>ESI 檔 : HIWINMIKROSYSTEM_ED1F_20250107 以上 |
| E2 系列 EtherCAT 驅動器 | 軟體 ( Thunder ) : 1.13.7.0 ( 含 ) 以上<br>韌體 : 3.13.6 ( 含 ) 以上<br>ESI 檔 : HIWINMIKROSYSTEM_ED2F_20250107 以上 |
| ACS SPiiPlusEC     | 軟體 ( SPiiPlus MMI Application Studio ) : 3.13.01 ( 含 ) 以上<br>韌體 : 3.13.01 ( 含 ) 以上                      |

# 目錄

|       |                              |      |
|-------|------------------------------|------|
| 1.    | 連線與模組設定.....                 | 1-1  |
| 1.1   | 建立新專案 .....                  | 1-2  |
| 1.2   | 載入 ESI 檔.....                | 1-5  |
| 1.3   | 搜尋連接裝置 .....                 | 1-6  |
| 2.    | 參數設定.....                    | 2-1  |
| 2.1   | 選擇軸與單位 .....                 | 2-2  |
| 2.2   | 設定裝置參數 .....                 | 2-4  |
| 2.3   | 下載參數設定至控制器 .....             | 2-8  |
| 3.    | 試運轉 .....                    | 3-1  |
| 3.1   | Motion Manager .....         | 3-2  |
| 3.1.1 | Jog Motion .....             | 3-3  |
| 3.1.2 | Relative Move .....          | 3-4  |
| 3.1.3 | Absolute Move.....           | 3-6  |
| 3.2   | 歸原點 .....                    | 3-8  |
| 3.2.1 | Program Manager .....        | 3-10 |
| 3.2.2 | Communication Terminal ..... | 3-12 |
| 3.2.3 | 執行歸原點.....                   | 3-13 |
| 4.    | 物件參數讀寫 .....                 | 4-1  |
| 4.1   | 讀取物件參數 .....                 | 4-2  |
| 4.2   | 寫入物件參數 .....                 | 4-4  |

# 1. 連線與模組設定

|     |               |     |
|-----|---------------|-----|
| 1.  | 連線與模組設定.....  | 1-1 |
| 1.1 | 建立新專案 .....   | 1-2 |
| 1.2 | 載入 ESI 檔..... | 1-5 |
| 1.3 | 搜尋連接裝置 .....  | 1-6 |

## 1.1 建立新專案

1. 開啟 SPiiPlus MMI Application Studio 並點擊 New workspace 圖示。




圖 1.1.1

2. 點擊 **Add Controller**，選擇 **Ethernet**，執行 **Refresh** 搜尋控制器。

註：控制器預設 IP 為 10.0.0.100。請正確設定電腦端的 IP 位址，確保裝置可正常識別。




圖 1.1.2




圖 1.1.3

## 3. 選擇控制器，點擊 **Connect**，即成功連線。




圖 1.1.4




圖 1.1.5

## 1.2 載入 ESI 檔

1. 右鍵點擊 **My Controller**，再依序選擇 **Add Component→Setup→System Setup**。




圖 1.2.1

2. 雙擊 **System Setup**，於下方 Properties 欄位取得需放置 ESI 檔的路徑。




圖 1.2.2

3. 開啟 Repository Data 路徑下的 EtherCAT 資料夾，並將 E 系列驅動器的 ESI 檔放置於此路徑。




圖 1.2.3

## 1.3 搜尋連接裝置

1. 雙擊 System Setup，點擊 Execute 進行連接裝置的組態掃描。




圖 1.3.1

2. 點擊 **Yes**，開始進行組態掃描，掃描過程如下圖。




圖 1.3.2

3. 組態掃描完成後，確認 Automatic Setup Task 視窗中的任務全部都是打勾狀態，再點擊 **OK**。正確辨識到裝置後，System Configuration 會顯示裝置名稱及軸號。

註：

- (1) 掃描組態的過程中若發生錯誤，請重新檢查裝置接線，再重新執行一次 System Setup。
- (2) 請確認控制器支援第三方 EtherCAT 驅動器，由 Communication Termial 執行#SI 確認。




圖 1.3.3




圖 1.3.4

## 2. 參數設定

|     |                  |     |
|-----|------------------|-----|
| 2.  | 參數設定 .....       | 2-1 |
| 2.1 | 選擇軸與單位 .....     | 2-2 |
| 2.2 | 設定裝置參數 .....     | 2-4 |
| 2.3 | 下載參數設定至控制器 ..... | 2-8 |

## 2.1 選擇軸與單位

1. 右鍵點擊 My Controller，再依序選擇 Add Component→Setup→Adjuster Wizard。




圖 2.1.1

2. 雙擊 Adjuster Wizard，選擇欲設定的軸。此以「Axis 0」為例。




圖 2.1.2

3. 點擊 **Axis Architecture**，依連接的馬達類型設定架構，並選擇使用者單位。

■ 線性馬達範例




圖 2.1.3

■ 旋轉馬達範例




圖 2.1.4

## 2.2 設定裝置參數

1. 於 Adjuster Wizard 視窗進入 Components 的 Motor 頁面，點擊編輯圖示，依馬達規格完成設定，再點擊右下角的 **Next >>**。此以「旋轉馬達 DMS03」為例。




圖 2.2.1

- 進入 Components 的 Drive 頁面，點擊編輯圖示，依驅動器電流規格完成設定，再點擊右下角的 **Next >>**。此以「E2-3A 驅動器」為例。




圖 2.2.2

- 進入 Components 的 Feedback 的 Motor 頁面，點擊編輯圖示，依驅動器的控制單位設定解析度，再點擊右下角的 **Next >>**。此以「DMS03 馬達、解析度 4325376 cnt/rev」為例。




圖 2.2.3

- 進入 Components 的 Calculate Parameters 頁面，執行 **Calculate Parameters**。參數計算完成後，再點擊 **Apply Changes** 將裝置參數導入設定。




圖 2.2.4




圖 2.2.5

- 進入 Miscellaneous Definition 的 Enable / Disable / Brake 頁面，依馬達類型設定 Enable Delay 時間。使用驅動器的 SW method 1 電機角檢出方法時，Enable Delay 建議設定為 2 秒(含)以上。

註：

- (1) 絕對式編碼器或霍爾感測器的電機角檢出方法不同，Enable Delay 時間可以維持預設值。
- (2) 線性馬達或直驅馬達需調整 Enable Delay 的設定，若未調整，激磁馬達時會觸發錯誤 ( Error 5019 )。




圖 2.2.6

6. 進入 Verification 的 Feedback 的 Motor 頁面，轉動馬達，確認位置回授與控制單位有匹配。此以「DMS03 馬達、解析度 4325376 cnt/rev」為例。




圖 2.2.7

## 2.3 下載參數設定至控制器

1. 於 Adjuster Wizard 視窗進入 Save to Flash 頁面，勾選設定好的軸，點擊 **Save** 將參數寫入控制器，再點擊右下角的 **Next >>**。




圖 2.3.1

2. 進入 Final Page 頁面，確認已完成 Adjuster Wizard 的設置，再點擊右下角的 **Finish**。




圖 2.3.2

## 3. 試運轉

|       |                              |      |
|-------|------------------------------|------|
| 3.    | 試運轉 .....                    | 3-1  |
| 3.1   | Motion Manager .....         | 3-2  |
| 3.1.1 | Jog Motion .....             | 3-3  |
| 3.1.2 | Relative Move .....          | 3-4  |
| 3.1.3 | Absolute Move .....          | 3-6  |
| 3.2   | 歸原點 .....                    | 3-8  |
| 3.2.1 | Program Manager .....        | 3-10 |
| 3.2.2 | Communication Terminal ..... | 3-12 |
| 3.2.3 | 執行歸原點 .....                  | 3-13 |

## 3.1 Motion Manager

1. 右鍵點擊 **My Controller**，再依序選擇 **Add Component**→**Diagnostics and Monitoring**→**Motion Manager**。




圖 3.1.1

2. 開啟 **Motion Manager** 視窗，依需求選擇運動指令。




圖 3.1.2

### 3.1.1 Jog Motion

1. 選擇移動軸，設定運動參數，點擊 **Enable**。此以「Axis 0、速度 60 度/秒」為例。




圖 3.1.1.1

2. 點擊 **Start Motion**，確認馬達以 60 度/秒的速度做 Jog 移動。




圖 3.1.1.2

## 3.1.2 Relative Move

1. 點擊 Create Motion，選擇 Relative Move。




圖 3.1.2.1

2. 選擇移動軸，設定運動參數，點擊 Enable。

此以「Axis 0、速度 60 度/秒、相對移動至位置 360 度」為例。




圖 3.1.2.2

3. 點擊 **Start Motion**，確認馬達以 60 度/秒的速度移動到相對位置 360 度。




圖 3.1.2.3

## 3.1.3 Absolute Move

1. 請參考 Relative Move 的步驟，點擊 **Create Motion**，選擇 **Absolute Move**。
  2. 選擇移動軸，設定運動參數，點擊 **Enable**。
- 此以「Axis 0、速度 60 度/秒、絕對移動至位置 360 度」為例。




圖 3.1.3.1

3. 點擊 **Start Motion**，確認馬達以 60 度/秒的速度移動到絕對位置 360 度。




圖 3.1.3.2

## 3.2 歸原點

歸原點需搭配 ACSPL+Programming 程序，請向本公司索取歸原點程序檔 ( Single axis homing example.prg ) 或複製以下程序於 Program Manager 的 Buffer 與 D-Buffer。

### ◆ Buffer

```
!Single-axis HIWIN drive homing process ver1.0
!!***** Setting Parameter *****!!
LOCAL INT X_HmMethod,X_HmType
LOCAL REAL X_HmVel,X_IndVel,X_HmAcc,X_HmOffset,X_CunitDmt,X_CunitNum,X_UnitType,X_Resolution
LOCAL INT ZONE0
AXISDEF X_axis=0
!!***** Operation Unit*****!!
X_UnitType = 1 ! 0 : Linear(mm). ; 1 : Rotary(deg).
X_CunitNum = 360 ! Enter the numerator of basic unit for Hiwin drive base on the X_UnitType parameter.
X_CunitDmt = 4325376 ! Enter the denominator of basic unit for Hiwin drive base on the X_UnitType parameter.
!!***** Operation Unit Setup example *****!!
!Linear motor with control unit 0.1 um. ***** !Rotary motor with control unit 0.0001 deg.
!X_UnitType = 0 ***** !X_UnitType = 1
!X_CunitNum = 1 ***** !X_CunitNum = 360
!X_CunitDmt = 10000 ***** !X_CunitDmt = 3600000
!!***** Basic Settings *****!!
X_HmMethod = 33 ! Enter the supported homing method.
X_HmType = 0 ! 0 : HIWIN drive internal homing. ; 1 : Touch probe homing.
X_HmOffset = 0 ! Homing offset base on the X_UnitType parameter.
X_HmVel = 30 ! Homing Velocity for finding home sensor base on the X_UnitType parameter.
X_IndVel = 10 ! Homing Velocity for finding home position base on the X_UnitType parameter.
X_HmAcc = 300 ! Homing acceleration base on the X_UnitType parameter.
!!***** Main program *****!!
FCLEAR X_axis
CALL ECATInOutOffset
CALL X_DetectDriverError
MFLAGS(X_axis).#HOME=0
IF X_HmType=0 ! HIWIN drive internal homing.
    CALL InterHmProc
ELSE ! Touch probe homing.
    CALL TouchProbeHmProc
END
DISP "Disable X_axis error compensation."
ERRORUNMAP X_axis, ZONE0
CALL RstMapping
ENABLE X_axis
TILL MST(X_axis).#ENABLED
WAIT 500
DISP "Axis is moving to home position....."
PTP/EV X_axis,0,X_HmVel
MFLAGS(X_axis).#HOME=1
FDEF(X_axis).#LL = 1
FDEF(X_axis).#SRL = 1
FDEF(X_axis).#SLL = 1
DISP "Axis Homing successful!!"
STOP
!!***** InOut mapping *****!!
ECATInOutOffset:
IF X_HmType = 0
    ECIN (ECGETOFFSET("Statusword",X_axis),X_StatusWord)
    ECIN (ECGETOFFSET("Error code",X_axis),X_ErrorCode)
    ECIN (ECGETOFFSET("Mode of operation display",X_axis),X_ModeOperDisp)
    ECOUT (ECGETOFFSET("Controlword",X_axis),X_ControlWord)
    ECOUT (ECGETOFFSET("Mode of operation",X_axis),X_ModeOper)
ELSE
    ECOUT (ECGETOFFSET("Touch probe function",X_axis),X_TouchProbFun)
END
RET
STOP
RstMapping:
ECUNMAPIN (ECGETOFFSET("Statusword",X_axis))
ECUNMAPIN (ECGETOFFSET("Error code",X_axis))
ECUNMAPIN (ECGETOFFSET("Mode of operation display",X_axis))
ECUNMAPOUT (ECGETOFFSET("Controlword",X_axis))
ECUNMAPOUT (ECGETOFFSET("Mode of operation",X_axis))
ECUNMAPOUT (ECGETOFFSET("Touch probe function",X_axis))
RET
!!***** Drive internal homing process *****!!
InterHmProc:
DISABLE X_axis
TILL ^MST(X_axis).#ENABLED
WAIT 500
FDEF(X_axis).#LL = 0 ! Motor is not disable when left limit signal rising.
FDEF(X_axis).#SRL = 0 ! Disable software right limit response.
FDEF(X_axis).#SLL = 0 ! Disable software left limit response.
ERRORUNMAP X_axis, -1 ! Close all axis error compensation.
X_Resolution=1/(X_CunitNum/X_CunitDmt)
```

```

IF X_UnitType = 0
    DISP"Linear resolution is %g (cunit/mm)!!", X_Resolution;
ELSE
    DISP"Rotary resolution is %g (cunit/deg)!!", X_Resolution;
END
COEWRITE/1 (X_axis,0x6098,0,X_HmMethod)
COEWRITE/4 (X_axis,0x6099,1,X_HmVel/(X_CunitNum/X_CunitDmt))
COEWRITE/4 (X_axis,0x6099,2,X_IndVel/(X_CunitNum/X_CunitDmt))
COEWRITE/4 (X_axis,0x609A,0,X_HmAcc/(X_CunitNum/X_CunitDmt))
COEWRITE/4 (X_axis,0x607C,0,X_HmOffset/(X_CunitNum/X_CunitDmt))
X_ModeOper = 6
TILL X_ModeOperDisp=6
WAIT 100
X_ControlWord=6
TILL X_StatusWord.0=1
WAIT 600
X_ControlWord=7
TILL X_StatusWord.0=1 & X_StatusWord.1=1
X_ControlWord=15
TILL X_StatusWord.0=1 & X_StatusWord.1=1 & X_StatusWord.2=1
WAIT 100
X_ControlWord=31
TILL (X_StatusWord.12=1 & X_StatusWord.10=1)|X_StatusWord.13=1
IF X_StatusWord.13=1
    CALL X_HomeFail
END
SET FPOS(X_axis) = COEREAD/4 (X_axis,0x6064,0)*(X_CunitNum/X_CunitDmt)
X_ControlWord=0
X_ModeOper=8
TILL X_ModeOperDisp=8
CALL RstMapping
WAIT 500
DISABLE X_axis
TILL ^MST(X_axis).#ENABLED
RET
STOP
!!***** Touch probe homing process *****!!
!!Search for negative limit switch in reverse direction to find the index
TouchProbeHmProc:
DISABLE X_axis
TILL ^MST(X_axis).#ENABLED
WAIT 500
MFLAGS(X_axis).#OPEN = 0
ERRORUNMAP X_axis, -1
MFLAGS(X_axis).#GANTRY = 0
FDEF(X_axis).#SRL = 0
FDEF(X_axis).#SLL = 0
FDEF(X_axis).#LL = 0
ENABLE X_axis
TILL MST(X_axis).#ENABLED
WAIT 500
DISP"Start to find the left limit....."
JOG/V X_axis,-X_HmVel
TILL (COEREAD/4 (X_axis,0x60FD,0) & 1) = 1
HALT X_axis
DISP"Already found the left limit!!"
WAIT 100
DISP"Start to leave the left limit....."
JOG/V X_axis,X_IndVel
TILL (COEREAD/4 (X_axis,0x60FD,0) & 1) = 0
DISP"Already leave the left limit!!"
X_TouchProbFun = 0
WAIT 200
X_TouchProbFun = 21
WAIT 200
DISP"Start to find the index in positive direction....."
TILL (COEREAD/2 (X_axis,0x60B9,0) = 3)
HALT X_axis
TILL ^MST(X_axis).#MOVE
DISP"Already found the index!!"
SET FPOS(X_axis) = (COEREAD/4 (X_axis,0x6064,0) - COEREAD/4 (X_axis,0x60BA,0))*(X_CunitNum/X_CunitDmt) + X_HmOffset
RET
STOP
!!***** Error detection *****!!
X_DetectDriverError:
IF X_ErrorCode<0
    DISP "Drive has error %x , Please check the driver",X_ErrorCode
    CALL X_HomeFail
END
RET
STOP
X_HomeFail:
DISP"Homing process fail. "
HALT X_axis
X_ControlWord=0
WAIT 100
X_ModeOper = 8
TILL X_ModeOperDisp=8
WAIT 100
CALL RstMapping
STOP

```

## ◆ D-Buffer

```
GLOBAL INT X_TouchProbFun,X_StatusWord,X_ControlWord,X_ModeOper,X_ModeOperDisp,X_ErrorCode
global int I(100),I0,I1,I2,I3,I4,I5,I6,I7,I8,I9,I10,I11,I12,I13,I14,I15,I16,I17,I18,I19,I20,I21,I22,I23,I24,I25,I26,I27,I28,I29
global real V(100),V0,V1,V2,V3,V4,V5,V6,V7,V8,V9,V10,V11,V12,V13,V14,V15,V16,V17,V18,V19,V20,V21,V22,V23,V24,V25,V26,V27,V28,V29
```

## 3.2.1 Program Manager

1. 右鍵點擊 My Controller · 再依序選擇 Add Component→Application Development→Program Manager。




圖 3.2.1.1

2. 點擊 Program Manager 的 Load from File 圖示，選擇歸原點程序檔後，點擊 Load。




圖 3.2.1.2




圖 3.2.1.3

3. 點擊 **OK**，載入歸原點程序檔。




圖 3.2.1.4

## 3.2.2 Communication Terminal

1. 右鍵點擊 **My Controller (ECH36215C)**，再依序選擇 **Add Component**→**Utilities**→**Communication Terminal**。




圖 3.2.2.1

2. 成功開啟 Communication Terminal 後，將 Show Messages 調整為 **On**。




圖 3.2.2.2

### 3.2.3 執行歸原點

1. 點擊 Program Manager 的 Buffer 圖示，完成歸原點參數設定並執行編譯。

此以「DMS03 馬達、解析度 4325376 cnt/rev、驅動器內部歸原點方法 33」為例。

註：

- (1) 請正確設定 X\_UnitType、X\_CunitNum、X\_CunitDmt、X\_HmMethod、X\_HmType，否則歸原點條件可能跟預期的運動條件不同。
- (2) 線性單位為 mm，旋轉單位為 deg，請依此固定單位來設定運動條件。
- (3) 請以驅動器的控制單位來設定 X\_CunitNum、X\_CunitDmt，設定範例請參考歸原點程序。
- (4) X\_HmType 為 0 代表驅動器內部歸原點；1 則代表 Touch probe 歸原點。
- (5) 請勿於 X\_HmMethod 輸入驅動器內部歸原點不支援的方法；Touch probe 歸原點範例固定為碰觸負極限後反向尋找 Index 位置。




圖 3.2.3.1

## 2. 點擊 Run 圖示，開始執行歸原點。




圖 3.2.3.2

## 3. 歸原點開始執行，由 Communication Terminal 確認歸原點成功且馬達確實回到原點位置。




圖 3.2.3.3

# 4. 物件參數讀寫

|     |              |     |
|-----|--------------|-----|
| 4.  | 物件參數讀寫 ..... | 4-1 |
| 4.1 | 讀取物件參數 ..... | 4-2 |
| 4.2 | 寫入物件參數 ..... | 4-4 |

## 4.1 讀取物件參數

EtherCAT 通訊物件可透過 ACSPL+Commands 的 COEREAD 指令讀取，詳細指令說明請參考 ACS 的 Knowledge Center。

1. 開啟 Program Manager 的 Buffer，輸入 COEREAD 指令即可讀取物件數值。

此以「讀取驅動器參數 Pt100」為例。

```
DISP "Pt100=%d",COEREAD/2 (0,0x2100,0);
STOP
```




圖 4.1.1

2. 執行編譯後再點擊 Run 圖示，即可於 Communication Terminal 顯示物件數值。




圖 4.1.2

## 4.2 寫入物件參數

EtherCAT 通訊物件可透過 ACSPL+Commands 的 COEWRITE 指令寫入，詳細指令說明請參考 ACS 的 Knowledge Center。

1. 開啟 Program Manager 的 Buffer，輸入 COEWRITE 指令即可改寫物件數值。

此以「改寫驅動器參數 Pt100」為例。

```
COEWRITE/2 (0,0x2100,0,1000)  
STOP
```




圖 4.2.1

- 執行編譯後再點擊 Run 圖示，確認參數 Pt100 數值有改寫正確，可搭配讀取物件參數指令做確認。




圖 4.2.2